



ImageJ/Fiji Macro Language

Elnaz Fazeli, Biomedicum Imaging Unit, University of Helsinki, Finland
Anna Klemm, SciLifeLab, NBIS, Sweden

2024-09-30


Preparations



Download material:

<https://tinyurl.com/ImageJMacro>

- Click on Code > Download ZIP
- unzip on your computer



Outline of This Session



- Biological Data Set and Image Analysis Problem
- How can we "talk" to Fiji? - Macro Recorder, Built-in Macro Function
- Step-by-Step Workflow

The Cell Atlas/Human Protein Atlas



Cell Atlas Aim:


Determine the subcellular location of all cellular proteins.

Experimental Methods:

- Antibody generation against 12.000 human proteins
- Immunostaining, 22 cell lines
- Automated confocal microscopy
- □ 82.152 images

Image Analysis Aim:

- Mapping 12.000 human proteins to 30 subcellular structures





Adapted from Thul, P.J. et al. (2017). A subcellular map of the human proteome. *Science* 356.

The Aim: Quantify Signal Accumulation Inside Nuclei Region



Image source: Human Protein Atlas

v19.proteinatlas.org/ENSG00000113368-LMNB1



Mean intensity of
green inside nucleus
region

Dataset:


- Subset of The Cell Atlas (Human Protein Atlas)
- 3 color stack: microtubules (magenta), protein detected by antibody (green), nuclei (blue)

The Aim: Quantify Signal Accumulation Inside Nuclei Region



Image source: Human Protein Atlas

v19.proteinatlas.org/ENSG00000113368-LMNB1



Mean intensity of
green inside nucleus
region

Results		
File Edit Font Results		
	Label	Mean
1	signal:0001-0041	26957.406
2	signal:0002-0291	20013.618
3	signal:0003-0320	38092.890
4	signal:0004-0670	18741.716
5	signal:0005-0696	19940.679
6	signal:0006-0677	16445.010
7	signal:0007-1198	20677.366
8	signal:0008-1168	20005.914
9	signal:0009-1250	24444.675
10	signal:0010-1457	20037.129
11	signal:0011-1651	26454.839
12	signal:0012-1788	5380.207
13	signal:0013-1832	24655.655
14	signal:0014-1945	15790.756
15	signal:0015-1935	18773.019
16	signal:0016-2019	17444.867
17	signal:0017-2021	9947.993

Exercise: Find a Workflow



Task: Create selections (ROIs) around nuclei Create selections for each nucleus


- Try to do some preprocessing first to get a smooth segmentation
- Add ROIs to ROI manager and analyze

Image:

- CellAtlas_Subset/711_D6_1.tif

Image source: Human Protein Atlas

v19.proteinatlas.org/ENSG00000113368-LMNB1





We now want to automate this process



IMAGEJ MACRO LANGUAGE

Recorder



Open record window:


- Plugins > Macros > Record...
- Choose language – if needed



Let's try with only one command: Split Channels



Recorder



Script Editor



Recorder:

- Discover commands
- Window can be edited, copied, pasted, cut etc.

Script Editor:

- For building a script
- Has color-coding, code-completion, run-option etc.

How to start a new Script



Importance of recorder



- Discover commands
- Get arguments for specific functions

But also:

- Record and save your workflow – for documentation and reproducibility!

Exercise: Record the Workflow



Task:

- Open the recorder and record the discussed workflow.
- Discuss open questions within the group.

Image:

- Any image from CellAtlas_Subset

Exercise: Record the Workflow - Solution



Cleaned-up recorder:



Recorded.ijm

```
Recorder
Record: Macro Name: Macro.ijm Create ?
run("Split Channels");
selectImage("C1-711_D6_1.tif");
rename("nuclei");
selectImage("C2-711_D6_1.tif");
rename("signal");

selectImage("nuclei");
run("Median...", "radius=8");
setAutoThreshold("Huang dark no-reset");
setOption("BlackBackground", false);
run("Convert to Mask");
run("Fill Holes");

run("Analyze Particles...", "size=2000-Infinity exclude add");

run("Set Measurements...", "mean redirect=None decimal=3");
selectImage("signal");
roiManager("Select", 7);
run("Measure");
```

Normalise the data name

Filter nuclear image and make binary image

Retrieve nuclei boundaries

Measure

Exercise: Record the Workflow - Solution



Cleaned-up recorder:



Recorded.ijm

```
Recorder
Record: Macro Name: Macro.ijm Create ?
run("Split Channels");
selectImage("C1-711_D6_1.tif");
rename("nuclei");
selectImage("C2-711_D6_1.tif");
rename("signal");

selectImage("nuclei");
run("Median...", "radius=8");
setAutoThreshold("Huang dark no-reset");
setOption("BlackBackground", false);
run("Convert to Mask");
run("Fill Holes");

run("Analyze Particles...", "size=2000-Infinity exclude add");

run("Set Measurements...", "mean redirect=None decimal=3");
selectImage("signal");
roiManager("Select", 7);
run("Measure");
```

Normalise the data name



Programming Basics I

VARIABLES

Variables: definition

A screenshot of the ImageJ/Fiji Macro Language editor window. The window has a menu bar with 'File', 'Edit', 'Language', 'Templates', 'Run', 'Tools', and 'Tabs'. Below the menu bar is a tab labeled '*New_ijm'. The main text area contains four lines of code: '1 totalArea = 100;', '2 fileName = "wildtype.tif";', '3 description = "Launching the script...";', and '4 thereAreCells = true;'. The first line is highlighted in yellow. At the bottom of the window, there are buttons for 'Run', 'Kill', 'Show Errors', and 'Clear'.

- Can hold numbers or phrases/strings, but only one at a time
- Used whenever a value is used many times inside the script
- You define a variable by assigning it some content
- Variable name is on the left followed by an equal sign followed by the item (or items) being assigned
- Variable names can only start with characters

Numeric Variables: assignment



```
*New_ijm
File Edit Language Templates Run Tools Tabs
*New_ijm *New_ijm
1 totalArea = 0;
2 numPixelsCell1 = 154;
3 numPixelsCell2 = 190;
4 pixelSize = 0.350;
5 totalArea = (numPixelsCell1 + numPixelsCell2)*pixelSize;
6 print(totalArea);
Run Kill Show Errors Clear
Started New_ijm at Fri Mar 17 13:24:57 CET 2017
```

- Content of numeric variables can be modified using mathematical operations
- After an assignment, the previous content (if any) is forgotten
- Good practice is to use **d2s** (decimal to string) when printing numbers
e.g. `print(d2s(totalArea));`

String Variables: concatenation



```
string_concatenation.ijm
1 text1 = "a";
2 text2 = "Hello";
3 text3 = "Hello everybody!";
4 text4 = " ";
5
6 text5 = text1 + text2;
7 print(text5);
8 print(text1 + text2);
9 print(text2 + "world!");
10 print(text2 + text4 + "world!");
```

```
string_numbers_concatenation.ijm
1 number1 = "2";
2 number2 = "3";
3 print(number1+number2);
4
5 number3 = 2;
6 number4 = 3;
7 print(number3+number4);
8
9 text = "image";
10 print(text + number3);
```

```
*New_.ijm
File Edit Language Templates Run Tools Tabs
*New_.ijm
1 fileName = "wt007.tif";
2 print(fileName);
Run Kill Show Errors Clear
```

What about this?

Exercise: Record the Workflow - Solution



Cleaned-up recorder:



Recorded.ijm

```
Recorder
Record: Macro Name: Macro.ijm Create ?
run("Split Channels");
selectImage("C1-711_D6_1.tif");
rename("nuclei");
selectImage("C2-711_D6_1.tif");
rename("signal");

selectImage("nuclei");
run("Median...", "radius=8");
setAutoThreshold("Huang dark no-reset");
setOption("BlackBackground", false);
run("Convert to Mask");
run("Fill Holes");

run("Analyze Particles...", "size=2000-Infinity exclude add");

run("Set Measurements...", "mean redirect=None decimal=3");
selectImage("signal");
roiManager("Select", 7);
run("Measure");
```

Normalise the data name

Introduction of the next Exercise



- Define a string variable with the name of the image
- Build `selectImage()` using the variables

```
run("Split Channels");  
  
selectImage("C1-711_D6_1.tif");  
rename("nuclei");  
  
selectImage("C1-711_D6_1.tif");  
rename("signal");
```

Exercise: Usage of Variables



- Replace the **highlighted text** using the variable *title*.
- Check out slide 19 (String Variables: concatenation) for help.
- **start with file:** Step_00_UsingVariables.ijm

```
title = "711_D6_1.tif";  
  
run("Split Channels");  
selectImage("C1-711_D6_1.tif");  
rename("nuclei");  
  
selectImage("C2-711_D6_1.tif");  
rename("signal");
```

Solution: Usage of Variables



```
title = "711_D6_1.tif";  
  
run("Split Channels");  
selectImage("C1-" + title);  
rename("nuclei");  
  
selectImage("C2-" + title);  
rename("signal");
```

Step_00_UsingVariables_Solution.ijm

Structuring the Code



Technical point: *Structuring the code using comments*

- Comments are non-interpreted elements of code
- They help structure the code
- They help collaborators interpret the original analyst's intentions
- Comments are introduced either by `//` or surrounded by `/* */`:

```
//This is a short comment

/*
    This is a very long comment, spanning over
    multiple lines, allowing line breaks
*/
```



We now know what variables are but..

HOW TO GET THE NAME OF AN IMAGE AUTOMATICALLY?



Built-in Macro Functions



- Not everything is recorded. Much more functions can be found at:
<https://imagej.nih.gov/ij/developer/macro/functions.html>
- Tip: do a page-search (CTRL+F)

Step 1:

Normalise the data name



```
File Edit Language Templates Run Tools Window Options Help Step_01_SplitAndRename.ijm
> Step_01_SplitAndRename.ijm
7  */
8
9  //Step1: Getting image information + Normalise the data name
10 //get general information
11 title = getTitle();
12
13
14 //split channels and rename them
15 run("Split Channels");
16 selectImage("C1-" + title);
17 rename("nuclei");
18 selectImage("C2-" + title);
19 rename("signal");
20
21
22
23
24
25
26
27
28
29
30
```

Run Batch Kill ☐ REPL Show Errors Clear ▶

▶ Step_01_SplitAndRename.ijm

More about the Built-in Macro Functions



Stack.setChannel(1);

Function with input

getTitle();

Function with output;

nameOfMyImage = getTitle(); output is assigned to a variable

getDimensions(width, height, channels, slices, frames);

Output is assigned to variables within the brackets

Exercise: Built-in Macro Functions



Task 1: Catch-up with the script

- include the getTitle() function.
- **start with file:** Step_00_UsingVariables_Solution.ijm

Task 2: explore the built-in macro functions.

- open a new script, set language to IJ1 macro
- What happens when you run **getDimensions(channels, height, width, slices, frames)?**
- Use the print() function to explore the content of the variables channels and width.
- Check the usage of the getDimensions function either using code autocompletion or on the “built-in macro function” website.

Solution: Built-in Macro Functions



Task 1: Catch-up with the script

- title = getTitle()
- solution file: Step_01_SplitAndRename.ijm

Task 2: explore the built-in macro functions.





- What happens when you run
getDimensions(channels, height, width, slices, frames)?
- correct usage:
getDimensions(width, height, channels, slices, frames);

-

The script editor supports you with colors and error messages.




Color-coding in Script-Editor




Read the error messages!

Auto-completion helps to avoid common mistakes



Workflow: How will we tackle the problem ?



The recorded workflow



Cleaned-up recorder:



Exercises/Recorded.ijm

```
Recorder
Record: Macro Name: Macro.ijm Create ?
run("Split Channels");
selectImage("C1-711_D6_1.tif");
rename("nuclei");
selectImage("C2-711_D6_1.tif");
rename("signal");

selectImage("nuclei");
run("Median...", "radius=8");
setAutoThreshold("Huang dark no-reset");
setOption("BlackBackground", false);
run("Convert to Mask");
run("Fill Holes");

run("Analyze Particles...", "size=2000-Infinity exclude add");

run("Set Measurements...", "mean redirect=None decimal=3");
selectImage("signal");
roiManager("Select", 7);
run("Measure");
```



Filter nuclear image and
make binary image



Retrieve nuclei
boundaries

Exercise: Preprocessing, Asking for User-Input



Task:

- Insert the preprocessing and segmentation steps to your code (median filter, thresholding, fill holes)
- Use the `getNumber()` function to ask the user for a minimum size of the nuclei in pixel
- **start with file:** Step_01_SplitAndRename.ijm
- use recorded commands from Recorded.ijm

Solution: Preprocessing, Asking for User-Input



```
File Edit Language Templates Run Tools Window Options Help Step_02_03_Preprocess_Analy...
Step_02_03_Preprocess_AnalyzeParticles.ijm
19  rename("signal");
20
21
22  //Step2: Prefilter nuclear image and make binary image
23  selectImage("nuclei");
24  //preprocessing of the grayscale image
25  run("Median...", "radius=8");
26  //thresholding
27  setAutoThreshold("Huang dark");
28  setOption("BlackBackground", true);
29  run("Convert to Mask");
30  //postprocessing of binary image
31  run("Fill Holes");
32
33
34  //Step3: Retrieve the nuclei's boundaries
35  num = getNumber("minimum size", 2000 );
36  selectImage("nuclei");
37  run("Analyze Particles...", "size=" + num + "-Infinity add"); //add to ROI-Manager by running and
38
39
40
41
```



Step_02_03_Preprocess_AnalyzeParticles
.ijm



ImageJ Basics & Programming Basics

ROI MEASUREMENTS & RESULT SAVING

The recorded workflow



Cleaned-up recorder:



Recorded.ijm

Recorder

Record: Name:

```
run("Split Channels");
selectImage("C1-711_D6_1.tif");
rename("nuclei");
selectImage("C2-711_D6_1.tif");
rename("signal");

selectImage("nuclei");
run("Median...", "radius=8");
setAutoThreshold("Huang dark no-reset");
setOption("BlackBackground", false);
run("Convert to Mask");
run("Fill Holes");

run("Analyze Particles...", "size=2000-Infinity exclude add");

run("Set Measurements...", "mean redirect=None decimal=3");
selectImage("signal");
roiManager("Select", 7);
run("Measure");
```

Normalise the data name

Filter nuclear image and make binary image

Retrieve nuclei boundaries

Measure

Step 5:





Make the measurement and save the result



Hint:

If no ROI is selected in the ROI Manager, will measure **all ROIs**.

Commands: `roiManager("deselect")` and `roiManager("Measure")`





Empty ROI Manager and Empty Results window

Make sure to start with an empty ROI-manager and an empty Results table!

```
roiManager("reset");  
run("Clear Results");
```

```
Step_06_CleanStart.ijm  
File Edit Language Templates Run Tools Tabs Options  
Recorded.ijm Step_06_CleanStart.ijm Step_07_batchProcessing.ijm  
1 /*  
2  * NEUBIAS Academy  
3  * ImageJ/Fiji Macro Language  
4  * anna.klemm@it.uu.se - BioImage Informatics Facility @SciLifeLab  
5  * April 2020  
6  */  
7  
8 //clean-up to prepare for analysis  
9 roiManager("reset");  
10 run("Clear Results");  
11  
12 //Step1: Getting image information + Normalise the data name  
13 //get general information  
14 title = getTitle();  
15  
16
```

Result saving



```
48 //Step 5: Measure signal in nuclear envelope's boundaries and save the result
49 run("Set Measurements...", "mean display redirect=None decimal=3"); //define the
50 selectWindow("signal");
51 roiManager("deselect"); //ensures that no ROI is selected
52 roiManager("Measure"); //measures active ROI or - if no ROI is selected - all RO
53 // Save results
54 saveAs("results", "C:/Users/Anna/Desktop/Neubias_output/Results.csv");
55
```

Macro record yourself saving the Results table!

Exercise: Make Measurements and Save Results



Tasks: Add to script:

- Clean up ROI Manager at the beginning
- Make measurements
- Save measurements to a file
- **Start with file:** Step_04_ForLoop.ijm

```
roiManager("reset");  
run("Clear Results");
```

```
roiManager("deselect");  
roiManager("Measure");
```

```
saveAs("results", ... );
```

Optional Tasks:

- Make output filename reflect image name
- Make output directory a variable
- Let user choose output directory with dialog

Step 5: Measure



Solution

Step_04_Measure.ijm



EXTRA-STEPS



Programming Basics

FOR-LOOPS and Batch processing



Technical point

Loops

Definite loop

Indefinite loops

A priori

A posteriori

- Known number of iterations
- 3 arguments:
 - Initialisation
 - Condition for loop entry, as a boolean
 - Iteration

- Test performed **BEFORE** instructions are executed

- Instructions always executed at least once
- Test performed **AFTER** instructions have been executed

```
for(i=0; i<10; i++){  
    //Instruction 1  
  
    //Instruction 2  
  
    //Instruction 3  
}
```

```
i=0;  
while(i<10){  
    //Instruction 1  
    //Instruction 2  
    //Instruction 3  
  
    i++;  
}
```

```
i=0;  
do{  
    //Instruction 1  
    //Instruction 2  
    //Instruction 3  
  
    i++;  
} while(i<10)
```


Programming Basics: For-Loops



```
ForLoop_Example1.ijm
File Edit Language Templates Run Tools Tabs
ForLoop_Example1.ijm
1 //use i only as condition for the for-Loop
2 for (i=0; i<10; i++){
3     print("Neubias is great!");
4 }
5
6 //use i additionally for calculations
7 for (i=0; i<10; i++){
8     result = i * 10;
9     print(result);
10 }
11
Run Kill Show Errors Clear
```

```
*ForLoop_Example2.ijm (Running)
File Edit Language Templates Run Tools Tabs
ForLoop_Example1.ijm *New_.ijm *ForLoop_Example2.ijm
1 //different writing for i++
2 for (i=0; i<10; i+=1){
3     print(i);
4 }
5
6 //use a different increment
7 for (i=0; i<10; i+=2){
8     print(i);
9 }
10
11
Run Kill Show Errors Clear
```



ForLoop_Example1.ijm



ForLoop_Example2.ijm

For-Loops Exercise (Type along)




- Get below code running in a new script
- Find (at least four) **different ways** to modify the code below to print „Hello!“ **only once** *instead of* 10 times.
- Modify below code to print 10 lines saying: “My favorite number is 0”, “My favorite number is 1”, ... , “My favorite number is 9”.

```
for (i=0; i<10; i++){  
    print("Hello!");  
}
```



**For-Loop
over all
images**



Exercise: Batch-processing



Task:

- Use the template below to loop your code over all files in the input folder.

```
input_path = getDirectory("input files");
fileList = getFileList(input_path);

for (f=0; f<fileList.length; f++){
    // Clean-up to prepare for next image
    roiManager("reset");
    run("Close All");
    run("Clear Results");
    // Open next image
    open(input_path + fileList[f]);
    print(input_path + fileList[f]);
    // Rest of the code
    // (...)
    saveAs("results", "C:/Users/Anna/Desktop/"+title+"_results.xls");
}
```

SOLUTION: Batch-processing



Step_05_batchProcessing.ijm



Readability:

- Use meaningful variable names
- Assign variable at the top of a script if the variable is used widely, or as close to where it is used as possible
- Comment your code: for you and others


Performance:

- `setBatchMode(true);`

Reproducibility:

- Add Initialization code: close windows, reset roiManager, reset Results table ...
- Save quality control files, e.g. save the ROI manager
- Use file names that refer to the original files
- Save the parameters used with the other results.
- Save the macro itself or document its version
- Consider sharing your macro and parameters as Supplementary Information

Where to continue



Other resources:

imagej.net/Introduction_into_Macro_Programming

forum.image.sc

www.springer.com/gp/book/9783030763930

Chapter 3, ImageJ Macro Language
(free download)

Resources connected to this course



Youtube video:

<https://www.youtube.com/watch?v=o8tfkdcd3DA>

Material:

<https://github.com/ahklemm>

Image.sc forum thread:

<https://forum.image.sc/t/neubias-academy-home-interactive-course-imagej-fiji-macro-language-questions-and-answers/38678>

Comparable workflow, but in CellProfiler:

<https://www.youtube.com/watch?v=QrzHQLiIDKM>



Raw images (tif) were provided by The Human Protein Atlas.

<https://www.proteinatlas.org/humanproteome/cell>

Thul, P.J. et al. (2017). A subcellular map of the human proteome. Science 356.

Acknowledgements



Material made by **Anna Klemm**
Based on NEUBIAS material,
especially Fiji Macro Writing by **Fabrice Cordelières**
with more material from **Giovanni Cardone** and **Ofra Golani**